
Вступ

 Розроблювальний програмний продукт за назвою "Робота з відеоадаптером VGA. Скрінсервер", призначений для керування відеоадаптером VGA і дозволяє продемонструвати роботу скрінсервера.

 1. Підстава для розробки

 Розробка ведеться відповідно до навчального плану за курсом "Системне програмне забезпечення".

 Тема курсової роботи "Робота з відеоадаптером VGA. Скрінсервер".

 2. Призначення розробки

 Програмний продукт призначений для вивчення роботи з відеоадаптером VGA і      створення скрінсервера – «Падаючі зірки».

 3. Вимоги до програмного виробу

 3.1. Функціональні характеристики

 Програмний продукт повинний виконувати наступні функції:

 1. використовувати 13h режим VGA (320 x 200 at 256 colors);

 2. задавати колір і кількість виведених крапок на екран;

 3. забезпечувати рух крапок. 

 3.2. Надійність

 Розроблювальний програмний виріб не повинний бути піддано помилкам виникаючим при функціонуванні ПЭВМ: відмовлення в роботі не повинний приводити до значних втрат інформації.

3.3. Умови експлуатації

 Програмний продукт "Робота з відеоадаптером VGA. Скрінсервер", повинний безперебійно функціонувати в нормальних умовах експлуатації:

 температура навколишнього середовища від 5 С до 35 С;

 вібрації, зовнішні магнітні, радіаційні й електричні полючи не повинні перевищити норми;

 Для нормальної експлуатації системи необхідні початкові знання по експлуатації ПЭВМ і практичні навички роботи в середовищі NORTON COMANDER.

 3.4. Склад і параметр технічних засобів

Функціонування програми цілком забезпечується стандартною конфігурацією IBM сумісних персональних ЕОМ.

3.5. Інформаційна і програмна сумісність

Розроблювальний об'єкт функціонує в середовищі MS DOS,  повинний бути написаний мовою Pascal і використовувати можливості,  надані  середовищем програмування Turbo Pascal.  Розроблювальне програмне забезпечення повинне забезпечити роботу з відеоадаптером VGA.

4. Програмна документація

Для розроблювального виробу повинні бути складені  програмні документи:

      Специфікація;

      Технічне завдання;

      Пояснювальна записка;

      Опис програми;

      Текст програми.

5. Стадії й етапи розробки

    5.1. Вивчення технічної документації.

    5.2. Розробка структури програми.

    5.3. Розробка алгоритму  даної програми.

    5.4. Розробка текстів використовуваних функцій.

    5.5. Налагодження програми.

    5.6. Складання програмної документації.

6. Порядок контролю і приймання

Розроблене програмне забезпечення повинне бути випробуване на контрольному прикладі.

Висновок

Список використаних джерел

 ВСТУП

Система формування зображення є невід'ємною частиною ПЭВМ IBM PC.  Призначене для висновку на екран зображення створюється у відеопам'яті. У IBM PC як відеопам'ять використовується частина оперативної пам'яті.  Відеопроцесор, що входить до складу відеоконтролера, по умісту відеопам'яті постійно формує зображення на екрані дисплея. Оскільки відеопам'ять доступн для мікропроцесор, то швидкіст формуванн зображенн відповіда швидкіст пересиланн вміст пам'яті, що перевищувала швидкість передач даних при підключенні відеотерміналу через послідовний порт. Адреси відеопам'яті і її характеристик є стандартними для всіх ПЭВМ лінії IBM PC, а також сумісних з ними.

 ПЭВМ типу IBM PC може мати одну з чотирьох відеосистем, що використовують різні типи відеомоніторів і, отже, різні типи відеоконтролерів. Відеоконтролер першого типу називається Монохромним Адаптером (МА), він забезпечує роботу тільки монохромного відеомонітора в символьному режимі і не підтримує графічного режиму. Відеоконтролер другого типу називається Кольоровим Графічним Адаптером (CGA). За допомогою контролера CGA підключається кольоровий монітор, що може працювати в двох різних режимах. У символьному режимі мається можливість вибирати колір тла і колір символу з восьми можливих квітів, а також один із двох рівнів інтенсивності кольору для символу. У графічному режимі низького дозволу (640x200 растрових крапок) можна працювати тільки з одним кольором. Третій відеоконтролер називається Удосконаленим Графічним Адаптером (EGA), що підтримує такий же символьний режим, що і CGA, і багатобарвний графічний режим більш високого дозволу.

У даному курсовому проекті розглянута робота з відеорежимом VGA.

 1. ВІДЕОРЕЖИМИ ГРАФІЧНОГО АДАПТЕРА VGA

               Графічний адаптер VGA (Video Graphic Array) установлюється на моделях 50, 60 і 80 ПЭВМ серії PS/2 фірми IBM. Крім персональних комп'ютерів серії PS/2, адаптер VGA може використовуватися також і на ПЭВМ IBM PC XT/AT. З погляду функціональних можливостей і продуктивності, VGA є трохи поліпшеною версією графічного адаптера EGA. Незважаючи на це, VGA підтримує більш широкий спектр відеорежимів, особливо при використанні моніторів зі змінюваною робочою частотою.
Також як і графічний адаптер EGA, VGA у своєму складі містить кілька програмно-керованих компонентів: блок керування електронно-променевою трубкою (ЭЛТ), блок синхронізації, графічний контролер і пристрій керування атрибутами висновку. Кожний з цих компонентів адаптера керується програмно. Програми обслуговування VGA у складі базової системи введення/висновку (BIOS) доступні по перериванню 10h. Використання функції з номером 0 даного переривання дозволяє установити адаптер в один з 24 стандартних відеорежимів, підтримуваних BIOS.

 Кожен компонент VGA містить у своєму складі ряд регістрів, використовуваних для керування функціонуванням адаптера. Для кожного відеорежиму в програмі відеообслужування BIOS утримується відповідна таблиця значень регістрів відеоадаптера, у зв'язку з чим, у більшості випадків, для установки необхідного відеорежиму замість безпосереднього запису в регістри адаптера досить скористатися програмами з BIOS. У тих випадках, коли необхідно установити відеорежим не підтримуваний програмами BIOS, необхідно, у першу чергу, з'ясувати, які ж значення повинні бути поміщені в керуючі регістри адаптера для чого потрібно досить глибоке розуміння роботи відеосистеми ПЭВМ.

 Чому виникає необхідність у створенні додаткових відеорежимів? Звичайно це зв'язано з бажанням мати можливість одержання на екрані зображення з більш високим дозволом у графічному режимі або можливістю висновку на екран великих порцій даних у текстовому режимі в порівнянні зі стандартними відеорежимами, підтримуваними базовою системою введення/висновку. Деякі із широко розповсюджених програмних систем, таких, наприклад, як Microsoft Word або Lotus 1-2-3 використовують свої власні відеорежими.

 2. КЕРУВАННЯ ВІДЕОРЕЖИМАМИ

 Відеорежими характеризуються наступними параметрами:

 вертикальним дозволом (кількістю рядків растра на екрані);

 горизонтальним дозволом (кількістю символів або пікселів у рядку);

 представленням даних у буфері; атрибутами висновку (колір, мерехтіння і т.п.).

При програмуванні VGA велике місце займає керування горизонтальним і вертикальним дозволом зображення на екрані дисплея. Адаптер VGA має значно менші можливості керування представленням даних в екранній пам'яті й атрибутами висновку в порівнянні з можливостями керування дозволом екрана. З цієї причини, найбільш простим шляхом переходу до нестандартного відеорежиму є використання програм BIOS для установки деякого стандартного відеорежиму з наступною зміною значень декількох регістрів відеоадаптера. Сигнали керування відеомонітором Як горизонтальне, так і вертикальний дозвіл визначається послідовністю погоджених у часі вихідних сигналів графічного адаптера, керуючих рухом електронного променя ЭЛТ. Зрозуміло, що зображення на екрані не є статичним електронний промінь "малює" зображення на екрані рухаючи по рядках растра зверху вниз (див. мал. 1). Весь екран цілком обновляється від 50 до 70 разів у секунду в залежності від відеорежиму. 

У процесі висновку кожного рядка растра інтенсивність електронного променя змінюється в залежності від сигналів подаваних на вхід монітора відеоадаптером (у кольоровому моніторі для кожного первинного кольору використовується власний електронний промінь, що, у даному випадку, несуттєво). У ЭЛТ електронний промінь рухається з постійною швидкістю уздовж рядка растра і вниз з одного рядка растра на наступну. Для керування настанням моменту часу переходу лучачи від крайньої правої крапки рядка до крайньої лівої крапки наступного рядка (тобто для керування горизонтальним зворотним ходом лучачи) адаптер VGA генерує сигнал горизонтальної синхронізації. Сигнал вертикальної синхронізації використовується для керування переміщенням електронного променя від крайньої правої позиції в самому нижньому рядку растра в лівий верхній кут екрана (керування вертикальним зворотним ходом лучачи).

 Відеоадаптер VGA завжди програмується таким чином, щоб час, необхідне для висновку даних з відеобуфера завжди було меншим загальної кількості часу розгорнення одного кадру. Це дає можливість висновку на екран бордюру (overscan), що облямовує власне виведене зображення, що дозволяє центрувати зображення на екрані.

 Керувати параметрами сигналів розгорнення, генерируемых відеоадаптером VGA можна шляхом зміни значень відповідних регістрів блоку керування ЭЛТ у складі VGA. Значення, що поміщаються в ці регістри задають тривалість відрізків часу, обмірюваний у так званих символьних одиницях часу (character clock). Символьна одиниця це відрізок часу необхідний для висновку на екран 8 пікселів у графічному режимі VGA і 8 або 9 пікселів в алфавітно-цифрових режимах.

 Параметри рядкової (горизонтальної) розгорнення визначаються: загальною кількістю часу, затрачуваним на висновок рядка растра загальна тривалість рядка (horisontal total);

 тривалістю часу в символьних одиницях протягом якого відбувається відображення даних з відеобуфера при висновку на екран одного рядка растра тривалість ділянки відображення в рядку (horisontal displayed). Різниця між загальною тривалістю рядка і тривалістю ділянки відображення визначає розмір горизонтального бордюру;

 значення (у символьних одиницях), при якому починається сигнал горизонтальної синхронізації (horisontal sync).

 Параметри кадрової (вертикальної) розгорнення аналогічні параметрам рядкового розгорнення, однак час тут виміряється в рядках растра відрізках часу, затрачуваних для висновку одного рядка растра на екран і переходу до початку наступного рядка: загальна кількість рядків растра в кадрі розмір кадру (vertical total);

 кількість рядків у кадрі, використовуваних для висновку даних на екран кількість рядків даних у кадрі (vertical displayed).

 Різниця між розміром кадру і кількістю рядків даних у кадрі визначає величину вертикального бордюру;

 номер рядка растра в якій починається сигнал вертикальної синхронізації (vertical sync).

 Обмеження на значення тимчасових параметрів Для правильного завдання тимчасових характеристик сигналів керування генерацією зображення на екрані монітора необхідно при розрахунку значень, що поміщаються в регістри, враховувати базові частотні характеристики відеомонітора й адаптера, такі як частота висновку пікселів (dot rate), частота рядків (scan rate) і частота кадрів (vertical scan rate). 

ЧАСТОТА ВИСНОВКУ ПІКСЕЛІВ. Ця характеристика (її також часто називають смугою пропущення відеомонітора) визначає швидкість висновку пікселів. Для формування сигналів, що задають швидкість висновку пікселів використовується пьезокристаллический генератор. У VGA допускається використання одного з трьох таких генераторів, кожний з яких забезпечує формування сигналів заданої частоти. Два таких генератори включені до складу відеоадаптера VGA. Вони забезпечують формування сигналів з частотами 25.175 Мгц і 28.322 Мгц. Крім того, може використовуватися генератор, установлений на системній платі ПЭВМ PS/2 моделей 50, 60 і 80. 

ЧАСТОТА РЯДКІВ. Дана характеристика визначає кількість рядків растра формованих за одну секунду. Частота рядків залежить від кількості пікселів у рядку і дорівнює частці від розподілу частоти висновку пікселів на кількість пікселів у рядку. 

ЧАСТОТА КАДРІВ. Визначає скількох кадрів за одну секунду виводиться на екран монітора. Частота кадрів при програмуванні VGA, також як і частота рядків, не задається безпосередньо, а визначається кількістю рядків у кадрі. Для обчислення частоти кадрів досить розділити частоту рядків на розмір кадру, заданий у рядках растра. Програмування відеорежимів Після вибору значень тимчасових параметрів відеорежиму можна приступати до програмування відеоадаптера VGA, що містить у собі: програмування блоку керування ЭЛТ; програмування блоку синхронізації; завдання частоти генератора пікселів; завдання висоти символів (у рядках растра); модифікація необхідних перемінних BIOS. Доступ до регістрів VGA здійснюється через порти введення/висновку за допомогою застосування команд асемблера IN і OUT або, при розробці програми мовою високого рівня, шляхом використання спеціальних функцій, еквівалентних по своїх діях цим командам. Для доступу до програм відеообслужування BIOS використовується переривання з номером 16 (10h).

Блок керування ЭЛТ виконує велику частину дій по керуванню рядковим і кадровим розгорненням. У регістрах цього блоку задається тривалість подаваних на вхід монітора сигналів керування рядковим і кадровим розгорненням. У ньому також здійснюється синхронізація моментів формування зазначених сигналів з вибіркою даних з відеобуфера і їхньою обробкою перед висновком на екран.Для запису даних у регістр необхідно помістити номер регістра в порт уведення/висновку з адресою 3D4h, після чого зробити запис нового значення в порт 3D5h (Див. листинг програми 1).

 3. Програмна реалізація

При програмуванні блоку керування ЭЛТ можна застосувати ряд спеціальних прийомів. По-перше, можна робити запис в один 16-бітовий порт замість послідовного висновку в два 8-бітових порти для одержання тих самих результатів: 

;AL = номер регістра 

mov al,RegNumber 

; AH = нове значення регістра 

mov ah,RegValue 

mov dx,3D4h 

;Запис у порт 3D4h/3D5h 

out dx,ax 

Якщо використовується доступ до 8-бітових портів, обов'язково необхідно заборонити переривання. У противному випадку, між записами в порти може відбутися апаратне переривання, програма обробки якого модифікує регістри блоку керування ЭЛТ, після чого вихідна програма не зможе правильно функціонувати.

 У випадку програмування адаптера VGA для відеорежимів BIOS з номерами 7 або 0Fh замість портів уведення/висновку 3D4h і 3D5h варто використовувати порти з адресами 3B4h і 3B5h відповідно. Ці адреси портів збігаються з адресами аналогічних портів монохромного дисплейного адаптера (Monochrome Display Adapter MDA), що дозволяє встановлювати на тому самому комп'ютері крім відеоадаптера VGA ще одну кольорову графічну підсистему. На відміну від попередніх моделей відеоадаптерів фірми IBM, що керують регістри VGA можуть бути не тільки записані, але і прочитані, що дає можливість програмі запам'ятати поточний стан регістрів адаптера перед тим як модифікувати значення, що утримуються в них.

 Пристрій синхронізації адаптера VGA виконує ряд взаємозалежних функцій, включаючи синхронізацію висновку символів з роботою генератора пікселів. Один рядок растра символів може відображатися або за 8 або за 9 тактів генератора пікселів або, іншими словами, в алфавітно-цифровому режимі кожен символ може складатися з 8 або 9 пікселів по горизонталі в залежності від того, які значення будуть поміщені в керуючі регістри блоку синхронізації.

 За замовчуванням, при використанні VGA, кожен символ складається з 9 пікселів по горизонталі. У графічному режимі або в 350-рядковому алфавітно-цифровому режимі (режим сумісності з EGA), блок синхронізації програмується таким чином, що кожен символ займає 8 пікселів у рядку. Використання додаткового (дев'ятого) піксела дозволяє підвищити чіткість відображуваного на екрані тексту. Регістри блоку синхронізації доступні через порти введення/висновку з адресами 3C4h і 3C5h. При програмуванні цього блоку використовується той же підхід, що і при програмуванні блоку керування ЭЛТ, описаний раніше. Однак, тут маються і свої особливості. Якщо необхідно змінити частоту генератора пікселів або розмір символу по горизонталі, потрібно "запустити знову" (reset) блок синхронізації установивши біт 1 регістра перезапуску в 1 і назад. Біти 2 і 3 багатоцільові регістри висновку (miscellaneous output register) використовуються для керування частотою генератора пікселів (див. табл. 5). Для зміни частоти генератора необхідно прочитати вміст регістра 3CCh, модифікувати біти 2 і 3 отримані значення і записати результат у порт 3C2h. При виконанні таких дій необхідно перезапустити блок синхронізації подібно тому, як це було описано вище. За замовчуванням, програми відеообслуговування BIOS встановлюють адаптер VGA в алфавітно-цифровий режим з 25 текстовими рядками на екрані. У цьому режимі зображення на екрані складається з 400 рядків растра і, отже, кожен символ займає по висоті 16 растрових рядків. В алфавітно-цифрових режимах можна збільшити кількість текстових рядків, що з'являються на екрані, зменшивши число рядків растра символу. Біти 0 4 регістри 09h (регістр максимальної кількості рядків у символі) блоку керування ЭЛТ використовуються для завдання розміру символу по вертикалі. Сюди міститься значення, що повинне бути на 1 меншим кількості рядків у растрі символу. Таким чином, за замовчуванням, у цьому регістрі утримується значення 0Fh. . Хоча значення в регістрі кількості рядків у символі може бути змінене безпосередньо, переважніше скористатися для його модифікації програмами відеообслуговування BIOS. Програми BIOS надають для цього досить гнучкі засоби і, крім завдання розмірів символу, дозволяють одновремменно здійснити вибір відповідної таблиці графічних представлень символів: 

; AH = 11h (номер функції BIOS) ; 

AL = 12h (номер подфункции) 

mov ax,1112h 

mov bl,0

; виклик програми BIOS 

int 10h 

Приведена послідовність команд може бути використана для переходу до режиму висновку символів з розміром по горизонталі 8 пікселів. При цьому програмами відеообслуговування BIOS виробляється завантаження необхідної таблиці символів у генератор символів, модифікуються значення необхідних регістрів блоку керування ЭЛТ писується нове значення в область опису характеристик поточного режиму відеоадаптера.

 Висновок

У курсовому  проекті на тему "Робота з відеоадаптером VGA. Скрінсервер" розглянуті основні методи  роботи з відеоадаптером VGA.

На підставі даного матеріалу була розроблена програмний  продукт - скрінсервер.  За допомогою даного програмного продукту реалізовані наступні функції:
 1. використовувати 13h режим VGA (320 x 200 at 256 colors);

 2. задавати колір і кількість виведених крапок на екран;

 3. забезпечувати рух крапок. 

АНОТАЦІЯ

Приводиться опис основних процедур та функцій програми.

Програма використає наступні технічні засоби: мінімальний комплект IBM PC з обсягом пам'яті не менш 640 Кб, жорстким диском обсягом не менш 10 Мб і відео адаптером VGA.

Реалізація програмного продукту виконана в середі Turbo Pascal.

Обсяг вхідних програмного модуля 7.5 Кб, обсяг EXE файлу, що виконується 4.5 Кб.
Модуль функціонується у середі MS-DOS версії 3.3  і вище.

1. ОПИС ЛОГІЧНОЇ СТРУКТУРИ.
1.1. Основні перемінні

   NumStars = 300;            { число зірок}

   ZShl = 4;

   MinSpeed = -500;           { Ці constants визначають максимальну і мінімальну швидкість}

   MaxSpeed = 500;

   ScreenWords = 32000;       { 320 * 200 div 2 }

   NumCol = 256;

{ Маються 256 квітів, розділених  на 8 діапазонів по 8 різних квітів,

  У такий спосіб при дозволі 32 відтінків кожного кольору, у межах від 0 до 63}

   NumRanges = 8;

type

   RGBType = record

      R, G, B : byte;         { Палітра }

   end;

const

   ColorRanges : array[1..NumRanges] of RGBType = (

(R: 00; G: 00; B: 00), {Чорний}

      (R: 63; G: 63; B: 63), {Білий}

      (R: 30; G: 30; B: 63), {Синій}

      (R: 30; G: 63; B: 30), {Зелений}

      (R: 63; G: 30; B: 30), {Червоний}

      (R: 63; G: 30; B: 10), {Жовтогарячий}

      (R: 63; G: 63; B: 10), {Жовтий}

      (R: 55; G: 10; B: 63)); {Фіолетовий}

type

   StarType = record

      X, Y, Z : integer;      { Інформація щодо кожної зірки}

      Spd,                    { Швидкість зірки }

      Color : byte;           { Інформація про колір }

   end;

   PosType = record

      X, Y : integer;         { Інформація щодо кожної крапки }

   end;

var

   StarData : array[1..NumStars] of StarType;

   ClearData : array[1..NumStars] of PosType;

   PaletteData : array[0..NumCol - 1] of RGBType;

   Speed,

   XCenter, YCenter : integer;

1.2. Основні функції і процедури.

Програма містить наступні функції та процедури:

procedure ScrollStars; - Головна процедура. Забезпечує виклик основних функцій і процедур.

procedure InitGraphics; assembler; - Инициализирует 13h режим VGA.

procedure CloseGraphics; assembler; - Повертає систему назад у текстовий режим.

procedure SetVGAPalette(var Palette); assembler; - Використовує повний набір  VGA палітри засобами BIOS }

procedure PutPixel(X, Y, Color : word); assembler; - Помістити pixel у реальний екран.

function GetRandom : integer; - Повертає випадкове число в інтервалі: [-74, 75]. 

function GetRandomColor : byte; - Повертає випадкове число в інтервалі: 2..8 

procedure InitStars; - Ініціалізація позиції зірки.

procedure InitPalette; - Ініціалізація палітри.

procedure InitRange(Range, Index : word); - Ініціалізація  зазначеного діапазону квітів.

procedure HandleStars; - Обмежує швидкість руху зірки. Обчислює нову позицію. Стирає зірку. Малює нову зірку.

2. Результати виконання програми.

Дана програма  скрінсервер – “зорепад”. Використовується 13-ий  режим VGA (320 x 200 у 256 квітах). При виконанні відбувається рух зірок на екрані монітора в різні сторони. У такий спосіб спостерігається  реальний зміст глибини і швидкості.

МIНIСТЕРСТВО ОСВIТИ УКРАЇНИ

НАЦIОНАЛЬНИЙ ТЕХНIЧНИЙ УНIВЕРСИТЕТ

“ХАРКIВСЬКИЙ ПОЛIТЕХНIЧНИЙ IНСТИТУТ”

КАФЕДРА ОБЧИСЛЮВАЛЬНОI ТЕХНIКИ ТА ПРОГРАМУВАННЯ

                                    ЗАТВЕРДЖУЮ

                                  Завідуючий кафедрою ОТП

                                  ______________(Домнін Ф.А.)

                                  "_____"_____________ 2000 р.

РОБОТА З ВІДЕОАДАПТЕРОМ VGA. СКРІНСЕРВЕР.

Текст програми

АП67.096192.021 12 00-1-1

                                   Розробники:





                                   Керівник проекту:




                                   ________________( Нiкiтiн Н.I.)

                                     "____"_____________2000 р.


                                   Виконавець:





                                   _____________( Шкурлапов I.М.)

"____"_____________2000 р.


Харків 2000

ЗАТВЕРДЖЕНО

АП67.096192.021 12 00-1-1 ЛЗ

РОБОТА З ВІДЕОАДАПТЕРОМ VGA. СКРІНСЕРВЕР.

Текст програми

АП67.096192.021 12 00-1-1 ЛЗ

Аркушів 10

Харків 2000

{   This is a star scroller in Turbo Pascal 7.0, using VGA mode 13h

   (320 x 200 at 256 colors).

   It has colored stars. The colors fade from black to bright if the stars

   move closer towards the viewer. The stars also move at different speeds,

   thus giving a real sense of depth and speed.

   While running, use the up and down arrow keys to change the speed.

   Press Esc to leave the program.

}

program Star_Scroller;

{$A+,B-,G+}

uses  Crt;

const

{ These constants may be modified. }

   NumStars = 300;            { Number of stars. }

   ZShl = 4;

   MinSpeed = -500;           { These constants define min and max speed. }

   MaxSpeed = 500;

{ This version has colored stars and stars moving at a different speed. If

  you want to turn off these features easily, use the following constants. }

   WhiteStars = False;

   SameSpeed = False;

{ These not. }

   ScreenWords = 32000;       { 320 * 200 div 2 }

   NumCol = 256;

{ There are 256 colors. I divided them into 8 ranges of 8 different colors,

  thus allowing 32 gradiations of each color, ranging from 0 to 63. With the

  following constants, the colors are defined. Black is specified to provide

  a black background and must have the first index. }

   NumRanges = 8;

type

   RGBType = record

      R, G, B : byte;         { Palette entry. }

   end;

const

   ColorRanges : array[1..NumRanges] of RGBType = (

      ( R : 00; G : 00; B : 00 ),    { Black }

      ( R : 63; G : 63; B : 63 ),    { White }

      ( R : 30; G : 30; B : 63 ),    { Blue }

      ( R : 30; G : 63; B : 30 ),    { Green }

      ( R : 63; G : 30; B : 30 ),    { Red }

      ( R : 63; G : 30; B : 10 ),    { Orange }

      ( R : 63; G : 63; B : 10 ),    { Yellow }

      ( R : 55; G : 10; B : 63 ));   { Purple }

type

   StarType = record

      X, Y, Z : integer;      { Information on each star. }

      Spd,                    { Speed of the star. }

      Color : byte;           { Coloring information. }

   end;

   PosType = record

      X, Y : integer;         { Information on each point to be plotted. }

   end;

var

   StarData : array[1..NumStars] of StarType;

   ClearData : array[1..NumStars] of PosType;

   PaletteData : array[0..NumCol - 1] of RGBType;

   Speed,

   XCenter, YCenter : integer;

procedure InitGraphics; assembler;

{ Initializes VGA mode 13h. }

asm

   mov   ax, 0013h

   int   10h

end;

procedure CloseGraphics; assembler;

{ Puts system back in text mode. }

asm

   mov   ax, 0003h

   int   10h

end;

procedure SetVGAPalette(var Palette); assembler;

{ Set entire VGA palette using BIOS. }

asm

   mov   ax, 1012h

   xor   bx, bx

   mov   cx, NumCol

   les   dx, Palette

   int   10h

end;

procedure PutPixel(X, Y, Color : word); assembler;

{ Put a pixel on the real screen. }

asm

   { Clip. }

   mov   ax, Y

   or    ax, ax

   jb    @Exit

   cmp   ax, 200

   jae   @Exit

   mov   cx, X

   or    cx, cx

   jb    @Exit

   cmp   cx, 320

   jae   @Exit

   mov   bx, 0A000h

   mov   es, bx

   mov   bx, 320

   mul   bx

   add   ax, cx

   mov   di, ax

   mov   ax, Color

   stosb

@Exit:

end;

procedure LimitI(var Value : integer; Min, Max : integer); assembler;

asm

   les   di, Value

   mov   cx, es:[di]

   mov   ax, Min

   cmp   ax, cx

   jg    @Exit

   mov   ax, Max

   cmp   ax, cx

   jl    @Exit

   mov   ax, cx

@Exit:

   mov   es:[di], ax

end;

function GetRandom : integer;

(* Returns a random number: [-74, 75] \ { 0 } *)

var

   A : integer;

begin

   repeat

      A := Random(150) - 75;

   until A <> 0;

   GetRandom := A;

end;

function GetRandomColor : byte;

{ Returns a random value in the range 2..8 }

begin

   if WhiteStars

      then GetRandomColor := 2

      else GetRandomColor := Random(7) + 2;

end;

function GetRandomSpeed : byte;

{ Returns a random value in the range 1..8 }

begin

   if SameSpeed

      then GetRandomSpeed := 4

      else GetRandomSpeed := Random(8) + 1;

end;

procedure InitStars;

{ Initializes star positions. }

var

   A : word;

begin

   Randomize;

   for A := 1 to NumStars do

   with StarData[A] do

   begin

      X := GetRandom;

      Y := GetRandom;

      Z := 1 shl ZShl + Random(256 shl ZShl);

      if WhiteStars

         then Color := 2

         else Color := GetRandomColor;

      Spd := GetRandomSpeed;

   end;

end;

procedure InitPalette;

{ Initializes the palette. }

procedure InitRange(Range, Index : word);

{ Initializes a specified range of colors. }

var

   A, B : word;

begin

   A := Index;

   B := 0;

   while B < 32 do

   begin

      PaletteData[A].R := (ColorRanges[Range].R * B) shr 5;

      PaletteData[A].G := (ColorRanges[Range].G * B) shr 5;

      PaletteData[A].B := (ColorRanges[Range].B * B) shr 5;

      Inc(A);

      Inc(B);

   end;

end;

var

   Range, PalIdx : word;

begin

   PalIdx := 0;

   for Range := 1 to NumRanges do

   begin

      InitRange(Range, PalIdx);

      Inc(PalIdx, 32);

   end;

   SetVGAPalette(PaletteData);

end;

procedure HandleStars;

var

   StarX, StarY : integer;

   A, StarCol : word;

   StarSpd : integer;

begin

   for A := 1 to NumStars do

   begin

  { Limit speed. }

      LimitI(Speed, MinSpeed, MaxSpeed);

  { Calculate new position. }

      StarSpd := (Speed + StarData[A].Spd * Speed) div 8;

      StarData[A].Z := StarData[A].Z - StarSpd -

                       ((StarSpd shl 4) div (StarData[A].Z shr ZShl));

      if StarSpd >= 0 then

      begin

         if StarData[A].Z < 1 shl ZShl then

         with StarData[A] do

         begin

            X := GetRandom;

            Y := GetRandom;

            Z := 256 shl ZShl;

            if WhiteStars

               then Color := 2

               else Color := GetRandomColor;

            Spd := GetRandomSpeed;

         end;

      end else

      begin

         if StarData[A].Z > 256 shl ZShl then

         with StarData[A] do

         begin

            X := GetRandom;

            Y := GetRandom;

            Z := 1 shl ZShl;

            if WhiteStars

               then Color := 2

               else Color := GetRandomColor;

            Spd := GetRandomSpeed;

         end;

      end;

      StarX := XCenter - ((StarData[A].X shl 8) div

          (StarData[A].Z shr ZShl));

      StarY := YCenter - ((StarData[A].Y shl 8) div

          (StarData[A].Z shr ZShl));

      { Clear star. }

      PutPixel(ClearData[A].X, ClearData[A].Y, 0);

      { Draw new star. }

      StarCol := (StarData[A].Color shl 5);

      PutPixel(StarX, StarY, StarCol - (StarData[A].Z shr (ZShl + 3)));

      ClearData[A].X := StarX;

      ClearData[A].Y := StarY;

   end;

end;

procedure ScrollStars;

{ This is the main procedure. }

var

   KeyCode : char;

begin

   XCenter := 160;

   YCenter := 100;

   Speed := 0;

   repeat

      HandleStars;

      if KeyPressed then

      begin

         KeyCode := ReadKey;

         if KeyCode = #0 then

         begin

            KeyCode := ReadKey;

            case KeyCode of

               #72 : Inc(Speed);

               #80 : Dec(Speed);

               else KeyCode := #0;

            end;

         end;

      end;

      Delay(10);

   until KeyCode = #27;

end;

begin

   InitStars;

   InitGraphics;

   InitPalette;

   ScrollStars;

   CloseGraphics;

end.

